Saturday	EE 671: VLSI Design	Due on
Aug. 27, 2022	Assignment 2	Sep. 05, 2020

In this assignment, we'll study the CPL and CVSL gates.
To reduce the work you have to do, a few templates are provided:
The first one is for the switch matrix used in all CPL gates.

```
* Switch Matrix
.subckt swmat In1 In2 In3 In4 con conbar Out1 Out2
MN1 In1 con Out1 0 cmosn
+ L=0.18U W=0.24U AD = 86.4fF AS = 86.4fF PD = 1.2U PS = 1.2U
MN2 In2 conbar Out1 0 cmosn
+ L=0.18U W=0.24U AD = 86.4fF AS = 86.4fF PD = 1.2U PS = 1.2U
MN3 In3 con Out2 0 cmosn
+ L=0.18U W=0.24U AD = 86.4fF AS = 86.4fF PD = 1.2U PS = 1.2U
MN4 In4 conbar Out2 0 cmosn
+ L=0.18U W=0.24U AD = 86.4fF AS = 86.4fF PD = 1.2U PS = 1.2U
* Loads representing wiring capacitance
C1  Out1    0 50fF
C2  Out2    0 50fF
.ends
```

In this template, con and conbar are the pivot inputs. Notice that wiring capacitance is included in the subcircuit.

We shall use the same model files as the ones in assignment-1 for 180nm CMOS. So minimum channel length for transistors is 180nm, the minimum width is 240 nm and the supply voltage is 1.8V. You should use the inverter designed by you in assignment-1 for the noise-margin restoring inverters which follow the switch matrix.

The template below generates all input combinations for the CPL and CVSL gates.

```
.param Trep1= 40n
.param Trep2 = {Trep1/2.0}
.param Trf = {Trep1/20.0}
.param Tw1 = {Trep1/2.0 - Trf}
.param Tw2 = {Trep2/2.0 - Trf}
.param hival=1.6
.param loval=0.2
V1  A    0 DC 0 PULSE({loval} {hival} {Tw1} {Trf} {Trf} {Tw1} {Trep1})
V2  Abar 0 DC 0 PULSE({hival} {loval} {Tw1} {Trf} {Trf} {Tw1} {Trep1})
V3  B    0 DC 0 PULSE({loval} {hival} {Tw2} {Trf} {Trf} {Tw2} {Trep2})
V4  Bbar 0 DC 0 PULSE({hival} {loval} {Tw2} {Trf} {Trf} {Tw2} {Trep2})
.tran 1pS {3*Trep1} 0nS
```

The parameter Trep1 can be adjusted to change the repetition rate of input pulses. Other parameters are automatically evaluated from this. The template generates A, Abar, B and Bbar as signals which transition between 0.2V as the logic '0' and 1.6V as the logic '1'. This is suitable for a supply voltage of 1.8V.
Since the period of B is half that of the A signal, it will generate all logic combinations of A

and B. Simulate a trial circuit by adding the following control block to the above template to look at the wave forms generated for A, Abar, B and Bbar:

```
.control
run
plot V(A)+2 V(Abar)+2 V(B) V(Bbar)
.endc
```

Q–1 Simulate a 2 input XOR gate implemented in CPL logic style. Load the final XOR and XNOR outputs with load capacitances of 100+nn fF, where nn are the last two digits of your roll number.

a) For this part, do not use the leakage reducing pMOS feedback in your circuit. Plot the voltages at the output of the switch matrix and the final XOR and XNOR outputs using transient analysis to show functionality of the gate. Also, plot the current drawn from V_{DD} and evaluate its average over the simulation period. Explain the wave forms seen.

b) Now add minimum sized pMOS transistors as the leakage reduction feedback. Observe and explain the voltage and current waveforms.

c) Next, add an additional 2 ns to the initial wait time for the wave forms for Abar and Bbar, keeping everything else the same as in the part above. (This describes the non-ideal case where signals and their complements are not strictly simultaneous). Re-simulate the circuit and describe what you observe.

d) To see the pseudo nMOS effect when feedback is used, use an exaggerated width for the feedback pMOS (say 4 times the minimum width) and see what happens. (Don't put the extra delay for Abar and Bbar for this part).

Q–2 Simulate a 2 input XOR gate implemented in CVSL style using inputs as generated for the previous question. Use the same pMOS transistor sizes as used in the CMOS inverter designed by you in assignment-1 and nMOS sizes which are scaled by series parallel rules over the CMOS inverter. Plot the transient response for the outputs.

– P.T.O. for model file

```
.MODEL CMOSN NMOS LEVEL=8 VERSION=3.3.0
+TNOM    = 27               TOX     = 4.1E-9
+XJ      = 1E-7             NCH     = 2.3549E17      VTH0    = 0.3662473
+K1      = 0.5864999        K2      = 1.127266E-3    K3      = 1E-3
+K3B     = 0.0294061        W0      = 1E-7           NLX     = 1.630684E-7
+DVT0W   = 0                DVT1W   = 0              DVT2W   = 0
+DVT0    = 1.2064649        DVT1    = 0.4215486      DVT2    = 0.0197749
+U0      = 273.8094484      UA      = -1.40499E-9    UB      = 2.408323E-18
+UC      = 6.504826E-11     VSAT    = 1.355009E5     A0      = 2
+AGS     = 0.4449958        B0      = 1.901075E-7    B1      = 4.99995E-6
+KETA    = -0.0164863       A1      = 3.868769E-4    A2      = 0.4640272
+RDSW    = 123.3376355      PRWG    = 0.5            PRWB    = -0.197728
+WR      = 1                WINT    = 0              LINT    = 1.690044E-8
*+XL      = 0               XW      = -1E-8
+DWG     = -4.728719E-9     DWB     = -2.452411E-9   VOFF    = -0.0948017
+NFACTOR = 2.1860065        CIT     = 0              CDSC    = 2.4E-4
+CDSCD   = 0                CDSCB   = 0              ETA0    = 2.230928E-3
+ETAB    = 6.028975E-5      DSUB    = 0.0145467      PCLM    = 1.3822069
+PDIBLC1 = 0.1762787        PDIBLC2 = 1.66653E-3     PDIBLCB = -0.1
+DROUT   = 0.7694691        PSCBE1  = 8.91287E9      PSCBE2  = 7.349607E-9
+PVAG    = 1.685917E-3      DELTA   = 0.01           MOBMOD  = 1
*+RSH     = 6.7
+PRT     = 0                UTE     = -1.5           KT1     = -0.11
+KT1L    = 0                KT2     = 0.022          UA1     = 4.31E-9
+UB1     = -7.61E-18        UC1     = -5.6E-11       AT      = 3.3E4
+WL      = 0                WLN     = 1              WW      = 0
+WWN     = 1                WWL     = 0              LL      = 0
+LLN     = 1                LW      = 0              LWN     = 1
+LWL     = 0                CAPMOD  = 2
*+XPART   = 0.5
+CGDO    = 8.23E-10         CGSO    = 8.23E-10       CGBO    = 1E-12
+CJ      = 9.466429E-4      PB      = 0.8            MJ      = 0.3820266
+CJSW    = 2.608154E-10     PBSW    = 0.8            MJSW    = 0.102322
+CJSWG   = 3.3E-10          PBSWG   = 0.8            MJSWG   = 0.102322
+CF      = 0                PVTH0   = -2.199373E-3   PRDSW   = -0.9368961
+PK2     = 1.593254E-3      WKETA   = -2.880976E-3   LKETA   = 7.165078E-3
+PU0     = 6.777519         PUA     = 5.505418E-12   PUB     = 8.84133E-25
+PVSAT   = 2.006286E3       PETA0   = 1.003159E-4    PKETA   = -6.759277E-3
+NOIMOD=2.0E+00 NOIA=1.3182567385564E+19
+NOIB=144543.977074592 NOIC=-1.24515784572817E-12 EF=0.92 EM=41000000
*
*
* flicker noise parameters above added manually from some other process
*
```

```
.MODEL CMOSP PMOS  LEVEL=8 VERSION=3.3.0
+TNOM    = 27              TOX     = 4.1E-9
+XJ      = 1E-7            NCH     = 4.1589E17      VTH0    = -0.3906012
+K1      = 0.5341312       K2      = 0.0395326      K3      = 0
+K3B     = 7.4916211       W0      = 1E-6           NLX     = 1.194072E-7
+DVT0W   = 0               DVT1W   = 0              DVT2W   = 0
+DVT0    = 0.5060555       DVT1    = 0.2423835      DVT2    = 0.1
+U0      = 115.6894042     UA      = 1.573746E-9    UB      = 1.874308E-21
+UC      = -1E-10          VSAT    = 1.130982E5     A0      = 1.9976555
+AGS     = 0.4186945       B0      = 1.949178E-7    B1      = 6.422908E-7
+KETA    = 0.0166345       A1      = 0.4749146      A2      = 0.300003
+RDSW    = 198.321294      PRWG    = 0.5            PRWB    = -0.4986647
+WR      = 1               WINT    = 0              LINT    = 2.94454E-8
+XL      = 0               XW      = -1E-8          DWG     = -2.798724E-8
+DWB     = -4.83797E-10    VOFF    = -0.095236      NFACTOR = 2
+CIT     = 0               CDSC    = 2.4E-4         CDSCD   = 0
+CDSCB   = 0               ETA0    = 1.035504E-3    ETAB    = -4.358398E-4
+DSUB    = 1.816555E-3     PCLM    = 1.3299898      PDIBLC1 = 1.766563E-3
+PDIBLC2 = 7.728395E-7     PDIBLCB = -1E-3          DROUT   = 1.011891E-3
+PSCBE1  = 4.872184E10     PSCBE2  = 5E-10          PVAG    = 0.0209921
+DELTA   = 0.01            RSH     = 7.7            MOBMOD  = 1
+PRT     = 0               UTE     = -1.5           KT1     = -0.11
+KT1L    = 0               KT2     = 0.022          UA1     = 4.31E-9
+UB1     = -7.61E-18       UC1     = -5.6E-11       AT      = 3.3E4
+WL      = 0               WLN     = 1              WW      = 0
+WWN     = 1               WWL     = 0              LL      = 0
+LLN     = 1               LW      = 0              LWN     = 1
+LWL     = 0               CAPMOD  = 2              XPART   = 0.5
+CGDO    = 6.35E-10        CGSO    = 6.35E-10       CGBO    = 1E-12
+CJ      = 1.144521E-3     PB      = 0.8468686      MJ      = 0.4099522
+CJSW    = 2.490749E-10    PBSW    = 0.8769118      MJSW    = 0.3478565
+CJSWG   = 4.22E-10        PBSWG   = 0.8769118      MJSWG   = 0.3478565
+CF      = 0               PVTH0   = 2.302018E-3    PRDSW   = 9.0575312
+PK2     = 1.821914E-3     WKETA   = 0.0222457      LKETA   = -1.495872E-3
+PU0     = -1.5580645      PUA     = -6.36889E-11   PUB     = 1E-21
+PVSAT   = 49.8420442      PETA0   = 2.827793E-5    PKETA   = -2.536564E-3
+ NOIMOD=2.0E+00           NOIA=3.574569933176E+18  NOIB=2500
+ NOIC=2.612600202858E-11  EF=1.1388                EM=41000000
*
*
* flicker noise parameters above added manually from some other process
*
```