EE671 VLSI Design - Assignment 3

Rohan Rajesh Kalbag, 20D170033

October 1, 2022

1 Approach

A subcircuit entity for the inverter from was taken from Assignment 1 to aid easier replication in the circuit. This inverter was designed such that the rise time and fall time are **233 ns** (200 + last two digits of roll number). The value of pMOS and nMOS widths taken were 1.505 μm and 0.502 μm. Thus we have the value of γ as $\frac{1.505}{0.502} = 2.998 \approx 3$.

Thus the value of γ for equal rise and fall times is 2.998 $\approx$ 3

1.1 Subcircuit for Inverter

```
*geometry parameters
.param pw = 1.505U
.param pl = 0.18U
.param pad = {2*pw*pl}
.param pas = {2*pw*pl}
.param ppd = {2*(pw + 2*pl)}
.param pps = {2*(pw + 2*pl)}

.param nw = 0.502U
.param nl = 0.18U
.param nad = {2*nw*nl}
.param nas = {2*nw*nl}
.param npd = {2*(nw + 2*nl)}
.param nps = {2*(nw + 2*nl)}
```

```
* unit inverter from assignment 1
.subckt inv supply Inp Output
MP1 Output Inp Supply Supply cmosp
+ L={pl} W={pw} AD={pad} AS={pas} PD={ppd} PS={pps}
MN1 Output Inp 0 0 cmosn
+ L={nl} W={nw} AD={nad} AS={nas} PD={npd} PS={nps}
.ends
```

In order to evaluate the delay for the inverter to measure the values of τ and p_{inv} the circuit provided in the problem statement was described in NGSPICE



Circuit Provided in Problem Statement

The template measure command provided was adjusted to get the rising (`delay_rising_edge`) and falling (`delay_falling_edge`) delays of the inverter using these two `.meas` commands. Then the average value of delay was taken and printed out for varying number of inverters as output load to the DUT as follows.

```
meas tran delay_rising_edge TRIG v(dutin) VAL=0.9 RISE=2 TARG v
   ↪ (dutout) VAL=0.9 FALL=2

meas tran delay_falling_edge TRIG v(dutin) VAL=0.9 FALL=2 TARG
   ↪ v(dutout) VAL=0.9 RISE=2

let delay = {(delay_rising_edge + delay_falling_edge)/2}

print delay
```

1.2 NGSPICE Netlist for Circuit

```
.include models-180nm

*geometry parameters
.param pw = 1.505U
.param pl = 0.18U
.param pad = {2*pw*pl}
.param pas = {2*pw*pl}
.param ppd = {2*(pw + 2*pl)}
.param pps = {2*(pw + 2*pl)}

.param nw = 0.502U
.param nl = 0.18U
.param nad = {2*nw*nl}
.param nas = {2*nw*nl}
.param npd = {2*(nw + 2*nl)}
.param nps = {2*(nw + 2*nl)}

* unit inverter from assignment 1
.subckt inv supply Inp Output
MP1 Output Inp Supply Supply cmosp
+ L={pl} W={pw} AD={pad} AS={pas} PD={ppd} PS={pps}
MN1 Output Inp 0 0 cmosn
+ L={nl} W={nw} AD={nad} AS={nas} PD={npd} PS={nps}
.ends

* pulse with time period of Trep, rise and fall times = Trep/20
.param Trep= 5n
.param Trf = {Trep/20.0}
.param Tw = {Trep/2.0 - Trf}
.param hival=1.8
.param loval=0.0
Vpulse pgen 0 DC 0 PULSE({loval} {hival} {Tw} {Trf} {Trf} {Tw}
   ↪ {Trep})

vdd supply 0 dc 1.8
```

```
x1 supply pgen n1 inv
x2 supply n1 n2 inv
x3 supply n2 dutin inv

*input shapers
xi1 supply n2 n3 inv
xi2 supply n2 n3 inv
xi3 supply n2 n3 inv
c1 n3 0 0.1p

*dut
xdut supply dutin dutout inv

*test load
xl1 supply dutout n4 inv
xl2 supply dutout n4 inv
xl3 supply dutout n4 inv
xl4 supply dutout n4 inv
xl5 supply dutout n4 inv
xl6 supply dutout n4 inv
xl7 supply dutout n4 inv
xl8 supply dutout n4 inv

*load on load
xll1 supply n4 n5 inv
xll2 supply n4 n5 inv
xll3 supply n4 n5 inv
xll4 supply n4 n5 inv
c2 n5 0 0.1p

.tran 0.1pS {3*Trep} 0nS
.control
run
meas tran delay_rising_edge TRIG v(dutin) VAL=0.9 RISE=2 TARG v
   ↪ (dutout) VAL=0.9 FALL=2
meas tran delay_falling_edge TRIG v(dutin) VAL=0.9 FALL=2 TARG
   ↪ v(dutout) VAL=0.9 RISE=2
let delay = {(delay_rising_edge + delay_falling_edge)/2}
```

```
print delay
let gamma = {1.505u/0.502u}
print gamma
.endc
.end
```

1.3 Results Obtained

Load Inverters (Fanout)	Average Delay (ps)	Rising Delay (ps)	Falling Delay (ps)
1	40.53	40.58	40.48
2	52.92	53.37	52.47
3	65.16	65.78	64.54
4	77.41	78.09	76.74
5	89.75	90.37	89.12
6	102.16	102.66	101.66
7	114.64	114.96	114.32
8	127.17	127.27	127.06



1.4 Inferences from Plot and Results

The line fitting the values was evaluated as **y** (delay in s) vs **x** (load fanout)

$$y = 1.244 \cdot 10^{-11}x + 2.912 \cdot 10^{-11}$$

Thus the value of τ would be the slope of this line (m) and the value of p_{inv} would be the y intercept (c).

τ is defined as the delay when one inverter drives another inverter ignoring self loading $\implies$ x = 1 and c = 0 which is the slope (m)

Thus the value of τ is given by $1.244 \cdot 10^{-11} \approx$ 12.44 ps

p_{inv} is defined as self loading delay of the inverter which is obtained when x = 0 which is the y-intercept

Thus the value of p_{inv} is given by $2.912 \cdot 10^{-11} \approx$ 29.12 ps

Also as shown earlier during importing the subcircuit for the inverter, the value of γ for which rise and fall times are equal is given by 2.998

Thus the value of γ for equal rise and fall times is 2.998 $\approx$ 3

1.5 Submission Details

The submission contains the `.xlsx` file having the readings obtained from the simulation which was used to plot the variation. It also contains the **NGSPICE** netlist `netlist.cir` used to simulate the circuit provided in the problem statement and this report `report.pdf`